/* Copyright (c) 2010 People Power Co.
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions
 * are met:
 * - Redistributions of source code must retain the above copyright
 *   notice, this list of conditions and the following disclaimer.
 * - Redistributions in binary form must reproduce the above copyright
 *   notice, this list of conditions and the following disclaimer in the
 *   documentation and/or other materials provided with the
 *   distribution.
 * - Neither the name of the People Power Corporation nor the names of
 *   its contributors may be used to endorse or promote products derived
 *   from this software without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
 * ``AS IS'' AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
 * LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS
 * FOR A PARTICULAR PURPOSE ARE DISCLAIMED.  IN NO EVENT SHALL THE
 * PEOPLE POWER CO. OR ITS CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT,
 * INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
 * (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR
 * SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION)
 * HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT,
 * STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED
 * OF THE POSSIBILITY OF SUCH DAMAGE
 *
 */


#include "ppp.h"
#include "HdlcFraming.h"

/** Top-level configuration for a PPP instance, not including LCP support.
 * @author Peter A. Bigot <pab@peoplepowerco.com>
 */
configuration PppC {
  uses {
    interface PppProtocol[ uint16_t protocol ];
    interface PppProtocolReject;
    interface HdlcUart;
    interface StdControl as UartControl;
  }
  provides {
    interface SplitControl;
    interface Ppp;
    interface PppRejectedProtocol;
    interface GetSetOptions<PppOptions_t> as PppOptions;
    interface GetSetOptions<HdlcFramingOptions_t> as HdlcFramingOptions;
  }
} implementation {
  components PppP;
  PppProtocol = PppP;
  PppProtocolReject = PppP;
  PppRejectedProtocol = PppP;
  SplitControl = PppP;
  Ppp = PppP;
  PppOptions = PppP;

  /** Maximum output frame comprises two octects for the protocol, and
   * the maximum supported MRU.  CRC octets are not stored in the
   * outgoing frame. */
  components new FragmentPoolC(2 + PPP_MAXIMUM_MRU, PPP_HDLC_TX_FRAME_LIMIT) as TransmitFramePoolC;
  PppP.TransmitFramePool -> TransmitFramePoolC;

  /* Maximum input frame comprises two octets for the protocol, the
   * maximum supported MRU, and two octets for the CRC.
   * Address/control field bytes are not stored. */
  components new HdlcFramingC(2 + PPP_MAXIMUM_MRU + 2, PPP_HDLC_RX_FRAME_LIMIT);
  PppP.HdlcFraming -> HdlcFramingC;
  PppP.HdlcControl -> HdlcFramingC;
  HdlcFramingOptions = HdlcFramingC;
  UartControl = HdlcFramingC;
  HdlcUart = HdlcFramingC;

  components LedC;
  PppP.MultiLed -> LedC;
}
